Exercise 3 (unit 2):

3. Calculate the fixed points of the functions $f(x) = (1 + a)\, x - b\, x^2$ and $g(x) = \frac{(1+a)\,x}{1+b\,x}$ and apply the convergence criterion. Relate your conclusions to the results obtained in Exercise 7 of Unit 1 (for simplicity, consider only the case a, $b > 0$).

```
Remove["Global`*"]
```

```
f = Function[x, (1 + a) x - b x^2];
Solve[f[x] == x, x]
f'[x]
```

```
{{x → 0}, {x → a/b}}
```

```
1 + a - 2 b x
```

We see that there are two fixed points, $x^* = 0$ and $x^* = \frac{a}{b}$.

```
f'[0]
```

```
1 + a
```

The first one is always a repellor, since $f'(0) = 1 + a > 1$.

```
f'[a / b]
```

```
1 - a
```

Moreover, $\frac{a}{b}$ is an attractor for the function f when $0 < a < 2$ (and the convergence is monotone when $0 < a < 1$). That is easily checked by hand, but we can also use *Mathematica:*

```
Reduce[Abs[1 - a] < 1, a, Reals]
Reduce[0 < 1 - a < 1, a, Reals]
```

```
0 < a < 2
```

```
0 < a < 1
```

Now let' s study the case $a = 2$.

```
a = 2;
b = 4;
Plot[{Nest[f, x, 10], Nest[f, x, 11], Nest[f, x, 12], Nest[f, x, 13]},
  {x, 0, 1}, PlotRange → {-1, 1}]
```




```
a = 2;
b = 3;
Plot[{Nest[f, x, 10], Nest[f, x, 11], Nest[f, x, 12], Nest[f, x, 13]},
 {x, 0, 1}, PlotRange → {-1, 1}]
```



It is an attractor and the convergence is oscillating. Now let' s study the case $a = 1$:

```
a = 1;
b = 4;
Plot[{Nest[f, x, 2], Nest[f, x, 3], Nest[f, x, 4], Nest[f, x, 5]},
 {x, 0, 1}, PlotRange → {-0.5, 0.5}]
```



The convergence is monotone.

Let’s study the function *g*.

```
Remove["Global`*"]
```

```
g = Function[x, (1 + a) x / (1 + b x)];
Solve[g[x] == x, x]
Simplify[g'[x]]
```

$$\left\{\{x \to 0\}, \left\{x \to \frac{a}{b}\right\}\right\}$$

$$\frac{1 + a}{(1 + b\,x)^2}$$

The function *g* has the same two fixed points, $x^* = 0$ and $x^* = \frac{a}{b}$.

```
g'[0]
```

```
1 + a
```

```
Simplify[g'[a / b]]
```

$$\frac{1}{1+a}$$

So $x^* = 0$ is always a repellor (since $g'(0) = 1 + a > 1$) and $x^* = \frac{a}{b}$ is always an attractor with monotone convergence (since $0 < g'\left(\frac{a}{b}\right) = \frac{1}{1+a} < 1$).